Example 3.5. A large construction firm is currently excavating at three sites. Meanwhile, they are also building at four additional sites, where they require fill dirt. The excavations at sites 1, 2, and 3 produce 150, 400, and 325 cubic yards of dirt per day. The building sites A, B, C, and D require 175, 125, 225, and 450 cubic yards of dirt per day. Additional fill dirt can also be obtained from site 4 at a cost of 5 dollars per cubic yard. The cost of shipping fill dirt is about 20 dollars per mile for one truckload, and a truck carries 10 cubic yards of dirt. Table 3.3 gives the distance between sites in miles. Find the transportation plan that minimizes the cost to the company.

Table 3.3

site	A	B	C	D
1	5	2	6	10
2	4	5	7	5
3	7	6	4	4
4	9	10	6	2

In [2]:
```
D=[5 2 6 10; 4 5 7 5; 7 6 4 4; 9 10 6 2]
```

Out[2]:
```
4×4 Matrix{Int64}:
 5   2  6  10
 4   5  7   5
 7   6  4   4
 9  10  6   2
```

In [3]:
```
C=2*D
C[4,:].+=5
C
```

Out[3]:
```
4×4 Matrix{Int64}:
 10   4  12  20
  8  10  14  10
 14  12   8   8
 23  25  17   9
```

In [4]:
```
using JuMP, HiGHS
```

In [5]:
```
model=Model(HiGHS.Optimizer)
```

Out[5]:
```
A JuMP Model
├ solver: HiGHS
├ objective_sense: FEASIBILITY_SENSE
├ num_variables: 0
├ num_constraints: 0
└ Names registered in the model: none
```

In [6]:
```
@variable(model,x[1:4,1:4].>=0)
```

Out[6]:
```
4×4 Matrix{VariableRef}:
 x[1,1]  x[1,2]  x[1,3]  x[1,4]
 x[2,1]  x[2,2]  x[2,3]  x[2,4]
 x[3,1]  x[3,2]  x[3,3]  x[3,4]
 x[4,1]  x[4,2]  x[4,3]  x[4,4]
```

In [7]:
```
cost(x)=sum(C.*x)
cost(x)
```

Out[7]:
$$ 10 x_{1,1} + 8 x_{2,1} + 14 x_{3,1} + 23 x_{4,1} + 4 x_{1,2} + 10 x_{2,2} + 12 x_{3,2} + 25 x_{4,2} + 12 x_{1,3} + 14 x_{2,3} + 8 x_{3,3} + 17 x_{4,3} + 20 x_{1,4} + 10 x_{2,4} + 8 x_{3,4} + 9 x_{4,4} $$

In [8]:
```
# Fix it so it doesn't scroll off the edge
println(cost(x))
```

```
10 x[1,1] + 8 x[2,1] + 14 x[3,1] + 23 x[4,1] + 4 x[1,2] + 10 x[2,2] + 12 x
[3,2] + 25 x[4,2] + 12 x[1,3] + 14 x[2,3] + 8 x[3,3] + 17 x[4,3] + 20 x[1,4]
+ 10 x[2,4] + 8 x[3,4] + 9 x[4,4]
```

In [9]:
```
@objective(model,Min,cost(x))
```

Out[9]:
$$ 10 x_{1,1} + 8 x_{2,1} + 14 x_{3,1} + 23 x_{4,1} + 4 x_{1,2} + 10 x_{2,2} + 12 x_{3,2} + 25 x_{4,2} + 12 x_{1,3} + 14 x_{2,3} + 8 x_{3,3} + 17 x_{4,3} + 20 x_{1,4} + 10 x_{2,4} + 8 x_{3,4} + 9 x_{4,4} $$

In [10]:
```
# Amount of dirt received by site j
r=[sum(x[:,j]) for j=1:4]
```

Out[10]:
```
4-element Vector{AffExpr}:
 x[1,1] + x[2,1] + x[3,1] + x[4,1]
 x[1,2] + x[2,2] + x[3,2] + x[4,2]
 x[1,3] + x[2,3] + x[3,3] + x[4,3]
 x[1,4] + x[2,4] + x[3,4] + x[4,4]
```

In [11]:
```
c1=@constraint(model,r.>=[175,125,225,450])
```

Out[11]:
```
4-element Vector{ConstraintRef{Model, MathOptInterface.ConstraintIndex{Math
OptInterface.ScalarAffineFunction{Float64}, MathOptInterface.GreaterThan{Fl
oat64}}, ScalarShape}}:
 x[1,1] + x[2,1] + x[3,1] + x[4,1] ≥ 175
 x[1,2] + x[2,2] + x[3,2] + x[4,2] ≥ 125
 x[1,3] + x[2,3] + x[3,3] + x[4,3] ≥ 225
 x[1,4] + x[2,4] + x[3,4] + x[4,4] ≥ 450
```

In [13]:
```
# Amount of dirt produced by site i
s=[sum(x[i,:]) for i=1:4]
```

Out[13]:
```
4-element Vector{AffExpr}:
 x[1,1] + x[1,2] + x[1,3] + x[1,4]
 x[2,1] + x[2,2] + x[2,3] + x[2,4]
 x[3,1] + x[3,2] + x[3,3] + x[3,4]
 x[4,1] + x[4,2] + x[4,3] + x[4,4]
```

In [14]:
```
c2=@constraint(model,s[1:3].<=[150,400,325])
```

Out[14]:
```
3-element Vector{ConstraintRef{Model, MathOptInterface.ConstraintIndex{Math
OptInterface.ScalarAffineFunction{Float64}, MathOptInterface.LessThan{Float
64}}, ScalarShape}}:
 x[1,1] + x[1,2] + x[1,3] + x[1,4] ≤ 150
 x[2,1] + x[2,2] + x[2,3] + x[2,4] ≤ 400
 x[3,1] + x[3,2] + x[3,3] + x[3,4] ≤ 325
```

In [16]:
```
print(model)
```

$$ \begin{aligned} \min\quad & 10 x_{1,1} + 8 x_{2,1} + 14 x_{3,1} + 23 x_{4,1} + 4 x_{1,2} + 10 x_{2,2} + 12 x_{3,2} + 25 x_{4,2} + 12 x_{1,3} + 14 x_{2,3} + 8 x_{3,3} + 17 x_{4,3} + 20 x_{1,4} + 10 x_{2,4} + 8 x_{3,4} + 9 x_{4,4}\\ \text{Subject to} \quad & x_{1,1} + x_{2,1} + x_{3,1} + x_{4,1} \geq 175\\ & x_{1,2} + x_{2,2} + x_{3,2} + x_{4,2} \geq 125\\ & x_{1,3} + x_{2,3} + x_{3,3} + x_{4,3} \geq 225\\ & x_{1,4} + x_{2,4} + x_{3,4} + x_{4,4} \geq 450\\ & x_{1,1} + x_{1,2} + x_{1,3} + x_{1,4} \leq 150\\ & x_{2,1} + x_{2,2} + x_{2,3} + x_{2,4} \leq 400\\ & x_{3,1} + x_{3,2} + x_{3,3} + x_{3,4} \leq 325\\ & x_{1,1} \geq 0\\ & x_{2,1} \geq 0\\ & x_{3,1} \geq 0\\ & x_{4,1} \geq 0\\ & x_{1,2} \geq 0\\ & x_{2,2} \geq 0\\ & x_{3,2} \geq 0\\ & x_{4,2} \geq 0\\ & x_{1,3} \geq 0\\ & x_{2,3} \geq 0\\ & x_{3,3} \geq 0\\ & x_{4,3} \geq 0\\ & x_{1,4} \geq 0\\ & x_{2,4} \geq 0\\ & x_{3,4} \geq 0\\ & x_{4,4} \geq 0\\ \end{aligned} $$

In [19]:
```
# Fix the scrolling left to right
print(sprint(print,model))
```

```
Min 10 x[1,1] + 8 x[2,1] + 14 x[3,1] + 23 x[4,1] + 4 x[1,2] + 10 x[2,2] + 12
x[3,2] + 25 x[4,2] + 12 x[1,3] + 14 x[2,3] + 8 x[3,3] + 17 x[4,3] + 20 x[1,
4] + 10 x[2,4] + 8 x[3,4] + 9 x[4,4]
Subject to
 x[1,1] + x[2,1] + x[3,1] + x[4,1] ≥ 175
 x[1,2] + x[2,2] + x[3,2] + x[4,2] ≥ 125
 x[1,3] + x[2,3] + x[3,3] + x[4,3] ≥ 225
 x[1,4] + x[2,4] + x[3,4] + x[4,4] ≥ 450
 x[1,1] + x[1,2] + x[1,3] + x[1,4] ≤ 150
 x[2,1] + x[2,2] + x[2,3] + x[2,4] ≤ 400
 x[3,1] + x[3,2] + x[3,3] + x[3,4] ≤ 325
 x[1,1] ≥ 0
 x[2,1] ≥ 0
 x[3,1] ≥ 0
 x[4,1] ≥ 0
 x[1,2] ≥ 0
 x[2,2] ≥ 0
 x[3,2] ≥ 0
 x[4,2] ≥ 0
 x[1,3] ≥ 0
 x[2,3] ≥ 0
 x[3,3] ≥ 0
 x[4,3] ≥ 0
 x[1,4] ≥ 0
 x[2,4] ≥ 0
 x[3,4] ≥ 0
 x[4,4] ≥ 0
```

In [20]:
```
optimize!(model)
```

```
Running HiGHS 1.13.1 (git hash: 1d267d97c): Copyright (c) 2026 under Apache
2.0 license terms
Using BLAS: blastrampoline
LP has 7 rows; 16 cols; 28 nonzeros
Coefficient ranges:
  Matrix  [1e+00, 1e+00]
  Cost    [4e+00, 2e+01]
  Bound   [0e+00, 0e+00]
  RHS     [1e+02, 4e+02]
Presolving model
7 rows, 16 cols, 28 nonzeros  0s
7 rows, 14 cols, 24 nonzeros  0s
Presolve reductions: rows 7(-0); columns 14(-2); nonzeros 24(-4)
Solving the presolved LP
Using dual simplex solver
  Iteration        Objective     Infeasibilities num(sum)
          0     0.0000000000e+00 Pr: 4(975) 0.0s
          5     7.6500000000e+03 Pr: 0(0) 0.0s

Performed postsolve
Solving the original LP from the solution after postsolve

Model status        : Optimal
Simplex   iterations: 5
Objective value     :  7.6500000000e+03
P-D objective error :  0.0000000000e+00
HiGHS run time      :          0.00
```

In [21]:
```
objective_value(model)
```

Out[21]:
```
7650.0
```

In [22]:
```
value(x)
```

Out[22]:
```
4×4 Matrix{Float64}:
   0.0  125.0    0.0    0.0
 175.0    0.0    0.0    0.0
   0.0    0.0  225.0  100.0
   0.0    0.0    0.0  350.0
```

In [ ]:
```

```